
Do you know the difference between global, private and local variables? What about PCOUNT() versus PARAMETERS() or PARAMETERS versus LPARAMETERS? Why should you declare variables and arrays? And what is a form property anyway, and why would you want to use one? What happens if you have a RETURN statement in the middle of a WITH…ENDWITH section? This session answers all these questions and more. Included are real world examples to help solidify these concepts in your mind.

What does “scope” mean?
“Scope” is the term used to refer to the period in which a variable is visible. For example, if you declare a variable in a function and try to use it in another function, it may not be available. We call that condition as being “out of scope.” If a variable is available to a function, then it is said to be “in scope.”

Other Terminology

Before we begin, we need to get some terminology straight. I use the following terms throughout this session.

Declare – To create a place for a variable in memory. This is different from initializing a variable, which is setting a variable’s initial value.
Function – A routine, function or method. This can be a method in a form, a function in a PRG or an entire PRG that has no FUNCTION command in it.
Various variables

The simplest discussion of scope involves the types of memory variables available. Visual FoxPro has three different types of variables, public, private and local and this how you specify the type of variable when you declare it:
PUBLIC goApp
&& Declares a public variable called goApp
PRIVATE pdToday
&& Declares a private variable called pdToday
LOCAL lnValue
&& Declares a local variable called lnValue
You can also declare multiple variables of the same type on the same line. For example:

LOCAL lnValue, lcReturn, llSuccess
&& Declares multiple variables at a time
Let’s cover the scope of each variable type.
Public
A public variable is visible until it is either released explicitly or the VFP session terminates. 
Public variables are automatically initialized to .F. when declared.
It is rarely good practice to use public variables. It could be acceptable to use a public variable to store a reference to a global application object near the beginning of your main program, if you release all variables when the app terminates. But a private variable can serve the same purpose and will release automatically when the app terminates. About the only legitimate purpose I can think of for a public variable is in some testing situations.

What’s wrong with using public variables?

The first thing wrong with using public variables is that you don’t always know they are public. If you’re at all sloppy about declaring variables or inconsistent in your naming convention, this can get you into big trouble. Here are a couple of examples:

PUBLIC ctest

cTest = "Test"

LOCAL cTest

cTest = "New Test"

DISPLAY MEMORY

The above code will display the following memory values:

CTEST     Pub    C    "Test"

CTEST     Local  C    "New Test"    test

If you query or set cTest, which one are you dealing with?  (Answer: The local variable takes precedence as long as it exists.)
Here’s another one:

PUBLIC ctest

cTest = "Test"

DO MyFunction

FUNCTION MyFunction

cTest = "New Test"

Let’s say you didn’t bother to declare cTest in MyFunction because you know that if you initialize a variable without declaring it first, it will be private. The problem is that you didn’t realize that this “private” variable was declared as public elsewhere, so what you’re changing is going to affect other parts of your app.
Another thing to keep in mind is that public variables stick around for the entire VFP session. That means that when you have a public variable in an app and terminate the app but not your VFP session (like when you’re in development), that variable still exists. When you start the app again, it may behave differently because the variable exists and has a value.
One of the best examples of how things can go horribly wrong when you use public variables was brought to my attention by Mike Yearwood.

Mike inherited an app developed by a team of programmers which used a public variable to track the current patient ID. Each form could change that variable. During testing everything seemed fine. It came as a surprise that records vanished. The fact is the records didn’t disappear. They moved to another patient. The records were a schedule for a nurse to give services to the patient. All it would have taken is a heparin injection to a hemophiliac and there could have been two deaths, one for the hemophiliac and one for lack of heparin. How many of us would like to be programmers on a system that caused deaths?
What caused the patient data to move to another patient? Let’s look at a simple scenario. You’re on a screen showing the services scheduled for this patient, and the patient ID is in a public variable. You then open a screen with information about the patient—address, date of birth, that kind of thing. But you need to look up something for a different patient, so you change the patient displayed on the patient information screen. That changes the value of the public variable. Then you return to the services screen—which is still showing the services scheduled for the original patient, even though the public variable has changed. You then make a minor change and save the data. Which patient does this get saved under? You guessed it—the wrong one!
Private
A private variable is visible in the function in which it is declared and any function called from the function in which it was declared. (Note that the value stored in a private variable by a called function may not be seen in the originating function if the variable was not initialized in the originating function.) A private variable is released as soon as the function in which it was created terminates. (Note that it can be created either by declaring it or by initializing it.)
Private variables are not automatically initialized by VFP.

If you create a variable by initializing its value without declaring it first, it will be private. (This is important to remember and is worth repeating.)
Local
A local variable is visible ONLY in the function in which it is declared. It will be released as soon as the function in which it was declared is terminated.
Local variables are automatically initialized to .F. when declared.
In most situations, local variables are the best choice. The reason for this is something called encapsulation. What is encapsulation? From http://www.wikipedia.com: “separation of concerns and information hiding, in software engineering, the process of enclosing programming elements inside larger, more abstract entities.”
Essentially, this means that a well encapsulated function is a black box. What the function uses internally is not visible to things outside other than the interface you have created to your function (i.e. the parameters the function accepts and the value it returns). Encapsulation reduces the chance of changes outside of your function breaking the function.
Naming conventions
It’s always a good idea to have a standard for naming variables. You should be consistent disciplined with your variables names so that you always use the standard you choose, whatever that standard may be.

Over the years, most VFP developers have adopted a standard referred to as “Hungarian Notation.” There is a topic in the VFP help file that discusses this under the topic, “naming conventions.” This topic has subtopics for a number of things, one of which is variables. This standard follows these rules:

The first character of a variable name represents its scope. The choices are:


l for LOCAL

p for PRIVATE


g for PUBLIC (or global as it’s called in other languages)

The second character of a variable name represents its type. Some examples:


c = character


n = numeric


d = date


t = datetime


o = object reference


a = array

These first two characters should always be shown in lower case.

The rest of the variable name should be something that is meaningful for its purpose. The first letter of this part of the name should be upper case to visually separate it from the first two characters. Use mixed case if you concatenate words together. Some examples:


pnGrandTotal


ldStartDate


llSuccess


ltModified


goApplication

VFP is very loose about variables. Although we have a standard for naming variables that shows their scope and type, VFP does not enforce variables to be any particular type or any particular size in memory. And although declaring variables before initializing them is always a good practice, it’s not necessary in VFP. For example if this is the first line in a function:

MyFirstVariable = "Some test value"
The line above both creates a variable called MyFirstVariable in memory and stores the value “Some test value” (without the quotes) into that variable. Remember, the default scope when you don’t specifically declare a variable is private.
Quiz
The following exercises should help your understanding of the scope of variables. (The answers to the quizzes are in Appendix A at the end of this paper.)
Exercise 1

Given this code:
FUNCTION Exercise1

LOCAL lnVal1, lnVal2

PRIVATE pnVal3, pnVal4

lnVal1 = 1

lnVal2 = 2

pnVal3 = .F.

DO Ex1a

? lnVal1

? lnVal2

? pnVal4

? pnVal5

ENDFUNC 

*------------------------------------------------------------

FUNCTION Ex1a

LOCAL lnVal1

lnVal1 = 3

lnVal2 = 4

pnVal3 = 5

pnVal4 = 6

ENDFUNC 

What will the display be for each of the variables?
Exercise 2

In the code below, the line poMyForm.Caption = "Test me" blows up with the error, “Object POMYFORM is not found.” Why?
FUNCTION Exercise2

DO Ex2CreateObj

poMyForm.Caption = "Test me"

ENDFUNC 

*------------------------------------------------------------

FUNCTION Ex2CreateObj

PRIVATE poMyForm

poMyForm = CREATEOBJECT("MyForm")

DEFINE CLASS MyForm as Form

ENDDEFINE 

ENDFUNC 
Arrays

An array is a list of variables. Arrays in VFP can be either one-dimensional or two-dimensional. Think of a one-dimensional array as a single column of values. A two-dimensional array has multiple columns as well as multiple rows. Arrays have scope just like variables do.
There are various ways to create an array. You can create an empty array with a command like PUBLIC, DIMENSION, DECLARE or LOCAL, in which you specify the number of rows and columns. You can create a populated array with a SQL SELECT command that sends its results to an array. You can also create and populate an array by calling any of the many VFP functions that send their results into an array of your choice.

Some examples:
PUBLIC ARRAY gaForms[5]
   && Creates a public array with 5 rows and 1 column
LOCAL ARRAY laResults[2,4] && Creates a local array with 2 rows and 4 column
DIMENSION paChoices[10]  && Creates a private array with 10 rows and 1 column
When declaring an array, you may use either square brackets as I’ve done in the examples above, or you may use parentheses. I prefer the square brackets because it more clearly distinguishes arrays from function calls, but you may have perfectly legitimate reasons to use parentheses instead.

Note that both the DIMENSION and DECLARE commands in VFP perform exactly the same function and both create a private array.
Note also that the ARRAY clause in the above examples is optional. You may declare arrays with only a LOCAL or PRIVATE command and you may do so in the same command in which you are declaring memory variables. I prefer to use the ARRAY clause, however, and to declare arrays in a separate command because if I forget to include the size of the array in the declaration, VFP will throw an error. If I omit the ARRAY clause in my declaration, VFP will assume the declaration is just a memory variable.

If you create an array as the result of a SQL query, it will be private unless you specify otherwise. For example:
SELECT * ;


FROM MyTable ;


WHERE llSomeCondition ;


INTO ARRAY laMyResults
The code shown above creates an array called laMyResults that is private, in spite of calling it laMyResults. If you want this array to be local, you must do this instead:

LOCAL ARRAY laMyResults[1]

SELECT * ;


FROM MyTable ;


WHERE llSomeCondition ;


INTO ARRAY laMyResults
I only have to create a 1 x 1 array when I declare it and VFP will re-dimension the array to match the size of the results.

The VFP functions that send results to an array work in a similar manner. If you don’t declare them first, they will be private—unless you’re in the Command Window, in which case they are public. Some examples of these functions are:

AFIELDS

ASESSIONS

AMEMBERS

APRINTERS

ADBOBJECTS
ACLASS

ALINES

ADATABASES
ASELOBJ

Parameterized Views

When defining a view that will accept a variable as part of the filter, you must decide if you want that variable to be a memory variable or a form property. There are pros and cons to each. 

If you choose to use a form property then you can refresh the view from any method of the form and the property will be in scope. You cannot, however, open that view outside of the form or from any other form that doesn’t have that property.
By choosing a variable, you simply have to initialize that variable and can then requery the view. The variable does not have to be in scope to browse, edit or update the view’s source. But this does mean that you must initialize the variable each and every time you open or requery the view. This also means that you can use that view anywhere, i.e. a Command Window or in any form.

Using a public variable as a parameter is not advisable for the same reasons already discussed. 

What I will often do for parameterized views is define them using a memory variable, but have a form property that stores the value I want for the entire scope of a form. Then I initialize the variable to the form property before re-querying the view.

One point worth mentioning is that when you precede the variable in your view definition with a question mark, that tells VFP to put up a dialog asking for the value of the variable if you don’t initialize it first. If you leave the question mark off and initialize the variable, the view works fine. If you forget to initialize the variable, however, VFP will throw an error.
Classes

In addition to variables, classes have properties and methods available. Properties are similar to variables and methods are similar to functions, except that they are scoped to the class. Many of these properties and methods are automatically created with the class. The specific properties and methods available depend on the type of class. For example, OptionGroups and Textboxes have a Value property but Labels and CommandButtons don’t. But classes may also have custom properties and methods. Think of a custom property as a variable you define that is private to the class. A custom method is a function that is private to the class. When you want the variable to be available in more than one method of a class, a class property is the way to go. If you need to call a function from a class method, but the function is too specialized for the class to be used outside of the class, a custom method is appropriate.
Creating properties and methods visually

Creation of class properties and methods is available when defining a class visually (VCX) in the Class Designer, as well as when defining a class in a PRG. To create a property or method when in the Class Designer, click on the Class menu pad and select the “New Property” or “New Method” option (Figure 1).

Figure 1 – The New Property option

After selecting “New Property”, the dialog shown in Figure 2 is displayed.

Figure 2 – Naming your new property
The only thing required on this dialog is the name of the property. We’ll cover the Visibility dropdown shortly. The other fields are beyond the scope of this session. Click the Add button to create the property and then click Close if you don’t need to create another property.

We must always preface a class property with THIS when we use it, so we don’t need to use the first character to indicate scope as we did with memory variables. But you should start with a letter that shows the property type, followed by a name to identify the purpose of the property.

If you need an array property for your form, you can create one by specifying the size of the array at the time you define the property’s name. See figure 3.

[image: image3.jpg]New Propertyetestvoamyctstomolass)

Mane:

Visbilly: | Publc
[accessMethod [ Assian Method

Defauil/Iniial Value:
t

Desciptior:

B B & B





Figure 3 – Creating an array property
Note that it doesn’t matter what size the array is when you create it. You can change it to be the correct size when you populate it later. As with other class properties, a class array property is scoped to the class. It will be released when the class is released.
Defining class properties in code

Just about anything that can be done in VFP visually can also be in code. You create a property for a class by simply setting the property to the initial value after the DEFINE CLASS command. The code below shows the creation of a ComboBox class, with a property called lNoDrop. I’ve highlighted the line that creates lNoDrop in yellow.
DEFINE CLASS cbobase AS combobox


Height = 24


Style = 2


Width = 112


DisabledForeColor = RGB(0,0,0)


BoundTo = .T.


*-- Determine whether to programmatically dropdown the list when the
* combo gets focus


lNoDrop = .T. 
&& Our custom property

Name = "cbobase"


PROCEDURE GotFocus


* Keyboarding Alt+DnArrow when the mouse is pressed causes the dropdown
* to drop and immediately pull back up.



IF This.Style = 2 AND NOT This.lNoDrop




IF mdown()





This.downclick()




ENDIF


ENDIF

ENDPROC

ENDDEFINE

You can also use the ADDPROPERTY function or method to add a property to a class at runtime.

Just because you can do something doesn’t mean you should. Writing visual classes in code can become extremely verbose! I don’t believe in using code to create visual objects. (Hey, it’s called VISUAL FoxPro for a reason!) I use the visual designers to create visual objects and code for classes that don’t have a visual component.

Scope of class methods and properties
There are three levels of scope available when a method (function) or property is defined in a class. These are:

· Public

· Protected

· Hidden

By default methods and properties are public. This means that anything that has access to the object reference to the class can call the method or access and set the property.
Protected methods and properties can only be seen by the class in which they were created and its subclasses.
Hidden methods and properties can only be seen by the class in which they were created.
When defining a class method or property, there is a dropdown from which you can select the scope. See Figure 4.

Figure 4 – Selecting scope visually
If you need to change a property or method already defined in a class, you can do that through the Class…Edit Property/Method menu option. See figure 5.

Figure 5 – Edit Property/Method menu option

This option displays the dialog shown in Figure 6.

Figure 6 – Editing a property or method

Note that all native VFP properties and methods are public by default. You can use the same dialog shown in Figure 6 to change the Visibility of native properties and methods just as you can for your own custom properties and methods. Doing so for a large number of properties or methods can become quite tedious and is not usually necessary.

When defining a class in code, you simply add the word PROTECTED or HIDDEN in front of the FUNCTION keyword to specify its scope. Declaring the scope of properties in code is similar to declaring variables in programs except that you have options to use PROTECTED and HIDDEN instead of PRIVATE and LOCAL.
DEFINE CLASS MyCustomClass as Custom


PROTECTED cMyProtectedVar


HIDDEN    cMyHiddenVar


cMyProtectedVar = "SomeValue"


cMyHiddenVar    = "SomethingElse"


cMyPublicVar    = "ThisIsPublic"


PROTECTED FUNCTION ClassNSubs



* Do something here that this class and



* its subclasses can call.


ENDFUNC 


HIDDEN FUNCTION ClassOnly



* Do something here that only this class



* can see.


ENDFUNC 

ENDDEFINE 

Forms

A form is nothing more than a specialized class. Like classes, you can create properties either visually or in code. Properties on forms can be accessed anywhere in the form by prefacing them with THISFORM. 

One difference with forms is that there are two visual tools for creating forms. You can create a form class through the Class Designer, which creates the form in a VCX file, or you can create an instance of a form class through the Form Designer, which creates the form in an SCX file. Creating methods and properties in either of these visual tools works the same way, except that you choose the Class menu pad when in the Class Designer and the Form menu pad when in the Form Designer. 

You cannot define the scope of methods and properties in SCX-based forms. They are always public, i.e. they are visible anywhere in the form and anything that can reference the form can reference the methods and properties of an SCX-based form.

Scope of Forms

One thing that confuses a lot of developers new to VFP is understanding the scope of forms. This varies depending on whether your form is a VCX-based form or SCX-based form as well as if it’s Modal (the form’s WindowType property is 1) or Modeless (the form’s WindowType property is 0).

SCX-based forms

SCX-based forms are run with a DO FORM command. In general, they stay in scope until you release them or your app terminates. (See the READ EVENTS section of this paper regarding termination of your app.) 

One complication to the DO FORM command is the optional NAME variable name LINKED clause. This clause will create a reference to your form that is stored in the variable name you specify. As with all objects, VFP will not release an object if there is a reference to it. This means that in order to release a form when there is a variable holding a reference to it, you must set that variable to NULL in order for the form to be released.

The NAME Clause

Just to complicate things, VFP also has a NAME clause that you can use with the DO FORM command. What this clause does is store an object reference to the form to a memory variable. This gives you the ability to reference the form from outside the form via the memory variable. But you must also remember to release the variable or set it to NULL when you release the form or you’ll have a “dangling reference” and the form won’t be released.

An example of how to use the NAME clause is:

LOCAL loMyForm

DO Form newform NAME loMyForm
You can now use loMyForm to reference all public items in this form, including methods, properties and objects on the form.

If you release loMyForm, the form still exists (provided there’s a wait state for it—see the Read Events section of this paper.) But you no longer have a way to reference it from outside the form.

Keep in mind that any variable you use with the NAME clause is subject to the same rules for scope as any memory variable. If the form you call is modeless, VFP will continue to run the code after the DO FORM command and the variable will go out of scope if it’s local.

VCX-based forms

VCX-based forms are classes that must be instantiated with the CREATEOBJECT() or NEWOBJECT() function. Doing so also required that you create a variable to hold a reference to this form object. If that variable should go out of scope, then your form object is released. Declaring a variable as PUBLIC for the purpose of keeping it in scope is this case is NOT a good practice. There are plenty of other ways of keeping the variable in scope that are much better.

PRG-based forms

PRG-based forms are very similar to VCX-based forms, however they are defined in code instead of the Class Designer. They are instantiated exactly the same way that a VCX-based form is.
Modal forms

Modal forms will stay in scope until you release them (and any references to them you might have created). For an SCX-based form, VFP will stop at your DO FORM command and not process the next line of code until the form is released. 

VCX-based forms can only be made modal by issuing a oForm.SHOW(1) command after instantiating the form (where oForm is the variable holding the reference to the form). This is necessary because objects instantiated with CREATEOBEJCT() or NEWOBJECT() aren’t visible until you make them so. If VFP stopped at the CREATEOBEJCT() or NEWOBJECT() command, your object would never be visible and the user would never be able to indicate it’s time to close the form.

Modeless forms

Modeless forms rely on a “wait state” to keep them in scope. VFP will continue processing commands in your program until it reaches this wait state. This means that if you are relying on a local variable as a reference to your form, that variable will go out of scope immediately. Please see the section on Read Events in this paper for more information about wait states.

The _Screen object reference
At this point I’m sure you’re wondering what kind of variables to use to store references to forms. I’ve already told you not to use public variables, so what choices do you have?
VFP provides a system variable that holds a reference to the main VFP screen called _Screen and one that holds a reference to the VFP session called _VFP. These variables are created automatically by VFP. The _Screen variable is available whether or not the VFP main screen is visible.

What this variable provides is a place where you can create properties that are available throughout your app without having to create your own application object. You would simply use the ADDPROPERTY method to create a new variable. For example:

_screen.AddProperty("MyVarible","Testing")
The above command creates a new property called MyVariable that belongs to the _screen object with an initial value of “Testing.” In order to read or change this value later in my app, I would use _Screen.MyVariable.

One difference between the two system variables worth noting is that _Screen has an AddProperty method, but _VFP does not. In order to add a property to the _VFP system variable, you must use the AddProperty function instead.

Why is the use of _VFP and _Screen better than using public variables? The most obvious is that it’s clear that you’re changing something that’s system-wide. Just by having to put _VFP or _Screen in front of your reference, it’s clear what the scope of the variable is. But if you’re running your app from the Command Window, these variables will persist after your app shuts down, which is not likely to happen if you create your own application object for storing things.

There’s another limitation to the use of these system variables. While they can store properties, you cannot add any methods to them. One of the common uses of an application object is to store methods used in multiple places in your app. You’re likely to find it necessary to create your own application object for this reason.

Use of a Form Manager

Many developers prefer to use a Form Manager. One of the purposes of a Form Manager is to keep object references to forms in scope. Form Managers generally will also keep track of the forms open and handle the opening and closing of forms for your app.

There are many examples of Form Mangers out there. Most frameworks come with a Form Manger built in. There is also one that comes with the book 1001 Things You Always Wanted to Know About Visual FoxPro, by Marcia Akins, Andy Kramek and Rick Schummer. (Available from www.hentzenwerke.com) 

READ EVENTS

Most apps need a READ EVENTS command somewhere in the main program. This is to provide a “wait state” for your app, or in other words, to keep it running until you issue a CLEAR EVENTS command. 

One of the most common questions I see on forums from new VFP developers goes something like this:

“My program just flashes on the screen and then disappears. It worked fine in development but not in production. What am I doing wrong?”
The answer 95% of the time is that the developer is missing a READ EVENTS clause in their program and therefore is not providing a place for the app to wait for the user.

All menus and modeless forms need a wait state to keep them alive. After displaying any menu or modeless form VFP continues processing lines of code until it reaches a wait state. A wait state can be a modal form or it can be a READ EVENTS command. If you have neither of these in your program, VFP just executes each line until the program ends, thereby terminating your VFP session. 

When in development, the VFP Command Window provides a wait state and this is why the app appears to work in development. A properly running program with its own wait state will not display the Command Window while running. If you see the Command Window while running your app, then you know you are missing a wait state.

A CLEAR EVENTS command will release this wait state so that your app can terminate. 
Quiz

Exercise 3

Given each of the following functions, which ones work and which ones don’t? Why? Describe the pros and cons of each of the techniques that work.
FUNCTION Ex3a

LOCAL loMyForm

loMyForm=NEWOBJECT("form1")

loMyForm.Show()

RETURN

DEFINE CLASS form1 AS form


ADD OBJECT cmdClose AS commandbutton WITH ;



Top = 96, ;



Left = 132, ;



Height = 27, ;



Width = 84, ;



Caption = "Close", ;



Name = "cmdClose"


PROCEDURE cmdClose.Click



Thisform.Release()


ENDPROC

ENDDEFINE

ENDFUNC 

*------------------------------------------------------------

FUNCTION Ex3b

LOCAL loMyForm

loMyForm=NEWOBJECT("form1")

loMyForm.Show(1)

RETURN

DEFINE CLASS form1 AS form


ADD OBJECT cmdClose AS commandbutton WITH ;



Top = 96, ;



Left = 132, ;



Height = 27, ;



Width = 84, ;



Caption = "Close", ;



Name = "cmdClose"


PROCEDURE cmdClose.Click



Thisform.Release()


ENDPROC

ENDDEFINE

ENDFUNC 

*------------------------------------------------------------

FUNCTION Ex3c

PUBLIC oMyForm

oMyForm=NEWOBJECT("form1")

oMyForm.Show()

RETURN

DEFINE CLASS form1 AS form


ADD OBJECT cmdClose AS commandbutton WITH ;



Top = 96, ;



Left = 132, ;



Height = 27, ;



Width = 84, ;



Caption = "Close", ;



Name = "cmdClose"


PROCEDURE cmdClose.Click



Thisform.Release()


ENDPROC

ENDDEFINE

ENDFUNC 

*------------------------------------------------------------

FUNCTION Ex3d

_Screen.AddProperty("oMyForm",NULL)

_Screen.oMyForm=NEWOBJECT("form1")

_Screen.oMyForm.Show()

RETURN

DEFINE CLASS form1 AS form


ADD OBJECT cmdClose AS commandbutton WITH ;



Top = 96, ;



Left = 132, ;



Height = 27, ;



Width = 84, ;



Caption = "Close", ;



Name = "cmdClose"


PROCEDURE cmdClose.Click



Thisform.Release()


ENDPROC

ENDDEFINE

ENDFUNC
*------------------------------------------------------------

This vs. ThisForm vs. Parent

Using THIS, THISFORM and THIS.PARENT allow you to reference other properties and methods in the container hierarchy. An example of container hierarchy is when you have a form with a pageframe and on one of those pages you have a grid. The grid has columns and each column has a header and a textbox. So, to specify the container hierarchy of the textbox, the textbox is contained within a column, which is contained within a grid which is contained within a page which is contained within a pageframe, which is contained within a form. In code you would specify this as something like oMyForm.Pageframe1.Page1.Grid1.Column1.Text1. As you can see, containership can get pretty deep! THIS, THISFORM and PARENT are all used to reference various levels of containership from within a form or class. 
 THIS refers to the current object and THISFORM refers to the form regardless of the object you’re in or its container hierarchy. THISFORM cannot be used outside of a form. If you are in a method of the form itself and not in the method of an object on the form, you can also access a form property or method by prefacing it with THIS. Using THIS in the method of an object on the form cannot be used to access a form property or method because in that case, THIS is referring to the object on the form rather than the form itself. You must always consider the context of where the code is when you use THIS.

If you are working with code that is in an object inside a container on a form—say a page of a pageframe—and you want to refer to the pageframe, you can use This.Parent. This is true for any object in a container, including grids, pageframes and the container class. A form is a container as well, so a command button on a form can reference the form with This.Parent. You can even go up multiple levels with multiple Parents. For example, when in the method of a page on a pageframe on a form, you can use This.Parent.Parent to reference the form. The use of Parent allows you to walk up the container hierarchy.

Parameters

There are a couple of different concepts of scope when discussing parameters. The first is in different ways to receive incoming parameters.

PARAMETERS vs. LPARAMETERS

When you have a function that receives parameters, you must declare those parameters in the first line of the function. Using a PARAMETERS statement makes those parameters PRIVATE. Using LPARAMETERS makes them LOCAL. In the code below, the line that displays tcName in the function Test1a will cause an error because tcName is not visible beyond the Test1function.
DO Test1 WITH "FirstName"

FUNCTION Test1

LPARAMETERS tcName as String

DO Test1a

ENDFUNC 

*---------------------------------------------

FUNCTION Test1a

? tcName

ENDFUNC
By changing the LPARAMETERS to PARAMETERS, the code will not cause an error because tcName is now Private and Test1a is called from Test1 so tcName is in scope.

PCOUNT() vs. PARAMETERS()

The PCOUNT() and PARAMETERS() functions illustrate the second concept of scope regarding parameters. Both of these functions are meant to tell you the number of parameters passed to a function. The difference is in their scope. As noted in the VFP 9 help file for the topic on PARAMETERS():
“The value returned by PARAMETERS() is reset every time a program, procedure, or user-defined function is called or when ON KEY LABEL is executed. Unlike PARAMETERS(), the PCOUNT() function does not get reset, so PCOUNT() may be preferable in most programming situations.”

What this means is that you can’t really rely on what PARAMETERS() returns to you. As an example, assume this code is in a PRG:
LPARAMETERS tcName

clear

? "Before: "+TRANSFORM(PARAMETERS())

DO OtherFunction

? "After: "+TRANSFORM(PARAMETERS())

RETURN 

*---------------------------------------

FUNCTION OtherFunction

* I don't do anything

ENDFUNC
The results displayed by this program are:

Before: 1

After: 0

But if you change this code to use PCOUNT() instead of PARAMETERS() the results are:

Before: 1

After: 1

It doesn’t seem likely that you would want to see the number of parameters change just because you’ve called another function, so in general it’s best to use PCOUNT() instead of PARAMETERS().

Passing by value vs. passing by reference

When a parameter is passed by value it means that the parameter passed from Function A is in a different location in memory from the parameter received by Function B. Therefore, changing the value of the parameter in Function B won’t affect the value of the variable in Function A.
But parameters may also be passed by reference. This means that instead of passing the value of a variable as a parameter, you are passing the reference to the location in memory. When you pass a parameter by reference, changing the value stored in that variable in Function B will change the value in Function A as well.

When you call a function with DO SomeProcedure WITH lcMyVar, the parameter is passed by reference.

When you call a function like SomeProcedure(lcMyVar), the parameter is passed by value.

It is possible to pass individual parameters to a function called as SomeProcedure(lcMyVar) by reference by preceding the variable with an @ symbol like so:

SomeProcedure(@lcMyVar)
You can also change this setting globally by using SET UDFPARMS TO REFERENCE and set it back again with SET UDFPARMS TO VALUE.

In order to pass a parameter by value when you use the DO…WITH syntax, you must enclose the individual parameters in parentheses like so:

DO SomeProcedure WITH (lcMyVar)

Arrays must always be passed by reference otherwise only the first element of the array is received in the function called.
Data sessions

We normally think of data sessions as something associated with a form. For now, that’s probably a good idea. Using a private data session on forms is best in most situations. It encapsulates the data being used by that form. 
One of the best examples of the advantage of using a private data session in a form that I can think of is when you have the same form opened twice by the same user. Say this form shows customer data and the user is working on one particular customer’s record. Then she gets a phone call from another customer that requires she look up that customer’s record. She can open a second instance of the same form and view the second customer’s data without having to close the form showing the first customer’s record and without affecting any record pointers on that first form.
Note that reports can also have a private data session. You would indicate a report has a private data session by checking the Private Data Session option on the Report menu pad. (See figure 7.)

Figure 7 – Setting a report to use a Private Data Session

Although I almost always use a private data session on forms, I never do in reports. The reason is that I don’t want to store preparation of data for a report inside of that report. This gives me the flexibility to use that data in output formats other than an FRX. For example, I might want the data shown in an Excel spreadsheet, use it in a Word document or simply export it to an ASCII file.

The trade-off of this technique is that it tightly couples a report with the form or program that generates the data for it. In other words, the report cannot run unless it’s called from something that will generate the data needed for the report. That function knows about everything the report needs. This dependence of the report upon something from outside and the knowledge of the form or program about the report is why we say the form and report are tightly coupled.

“Default data sessions”
The meaning of “default data session” is overloaded in VFP and can cause some confusion. When you first start up VFP, you are in data session 1 or the “default” data session. This is a different concept from a form with its DataSession property set to “1 – Default Data Session.” When a form uses a DataSession property of 1, it means it uses the same data session as whatever calls it. If that form is called from your main program, then that form uses data session 1. If that form is called from another form with a private data session of its own, then the second form uses the same datas ession as the first form, which is isolated from everything else in your application. 
DataEnvironment vs. Data Session
At this point I should really distinguish the difference between the terms data session and DataEnvironment. The difference is quite subtle. A DataEnvironment is a container that holds cursors, CursorAdapters and relations. A data session is an instantiation of a DataEnvironment, identified by a DataSessionID property.
Classes and data sessions 
There is an interesting ‘gotcha’ with objects and data sessions that is very important to know about. An object runs in the data session in which it was instantiated unless you specifically change it. This is particularly important when you have application methods that act on data. If you instantiate your application object from your main program, all methods in that object use the default data session (data session 1), which may not contain the cursors you expect if you call that method from a form with a private data session. You may need to pass the form’s DataSessionId to the method and have that method use SET DATASESSION TO in order to switch the data session used by the method.

Note that if your function is in a PRG that is not defined as a class, it will operate in the same data session as whatever called it. Therefore, you may prefer to keep any functions that operate on data out of your application object.

SET commands
There are many SET commands that are scoped to the data session. That means that if you set them in one data session and you change to another data session, those settings will be back at their defaults. See the VFP help topics, “SET DATASESSION Command” and “Commands that Scope to a Data Session” for a list of these.
Some commonly used SET commands that are scoped to the data session are:

SET TALK


SET CONFIRM

SET NULL

SET DELETED

SET EXCLUSIVE

SET ANSI

SET EXACT


SET SAFETY


SET CENTURY

SET MULTILOCKS

SET NEAR

Telling you to use private data sessions on your forms and then telling you that you have to redefine all your SET commands for each data session may seem like I’m making extra work for you, but there’s an easy solution. I simply put all my SET commands in a function in a PRG. I like to keep this outside of my application object so I don’t have to worry about switching data sessions. It doesn’t matter if these commands are scoped to the data session or not. I set all of them in one place. Then I call this function from any place that instantiates a new data session. For forms, that’s the Load method of my base form class. If I have views in a form’s DataEnvironment, then I might have to call my function from the data session’s BeforeOpenTables method. Otherwise SET TALK is ON and it might display unwanted results on the previous form. (This happens before the current form is instantiated, so any results are displayed on the previous form.)
Scope of functions

There is also an issue of what functions are available at any given time. Any function that is in the current PRG or any PRG that called the current PRG is in scope. This may lead to some name conflicts, as you may have functions with the same name in multiple PRGs. You may also have a PRG with the same name as a function inside of another PRG.
When you call a program or function with a DO command, VFP will first look for a function in the current PRG with that name. If it can’t find one, it will look in each PRG up the calling chain for the function. If it’s still not found, it will look for a standalone file with that name, i.e. an EXE, APP, FXP or PRG file.
If you do have multiple functions with the same name, and you know you want to call the one in the XYZ.PRG, you can simply say DO MyFunction IN XYZ. Note that I did not use the .PRG extension in that command. This is so that VFP can find in it whatever level of compiled file exists for XYZ. For example, in your test environment all your code may be in PRG files. But for production, you may compile each PRG into a separate APP file. Your PRG files wouldn’t exist in the production environment and if you used the .PRG with the IN clause, VFP would either not run the correct function or would fail.

When you call a PRG with the DO command, that program doesn’t have to do anything immediately. Just by calling it, you’ve brought any functions in that PRG into scope. This is an alternate technique to using SET PROCEDURE TO. I prefer SET PROCEDURE TO, however, and to have SET PROCEDURE TO near the top of my main program so I don’t have to dig into a lot of code to figure out what program files are called by my app.

If you have a mix of programs called with a DO command and in a SET PROCEDURE TO, anything in the SET PROCEDURE TO will take precedence over the chain of programs in the DO command.

Quiz

Exercise 4

You have the following code in FirstPRG.prg:

DO SecondPrg

DO Test
FUNCTION Test

MESSAGEBOX("Test function in FirstPrg")

ENDFUNC
Next is SecondPRG with this code:

SET PROCEDURE TO ThirdPRG

DO test
 And finally, ThirdPRG:

FUNCTION Test

MESSAGEBOX("Test in ThirdPRG")

ENDFUNC
What is displayed when you run FirstPrg.prg?

WITH…ENDWITH

Putting a series of commands between WITH and ENDWITH commands can be very useful. It can also be extremely dangerous if you don’t know what you’re doing. (I know of one developer who recommends that WITH…ENDWITH never be used at all.)
As an example, assume you have a grid on a form that may be setup in a couple of different ways, depending on a parameter sent to the form. Because parameters are received in a form’s Init, which is after the grid is instantiated, you may decide it’s best to setup this grid in code. You’ve created a custom property on the form in which you store the parameter and a method called GridSetup just for this purpose. It looks something like this:
* Method GridSetup

* Defines the number of columns in the grid, based on a parameter sent to the form.

* Possible parameters are: Full - show all columns

*                          Brief - show first 2 columns only

WITH Thisform.MyGrid as Grid

.RecordSource = "MyCursor"


DO CASE


CASE Thisform.cFormat = "Full"



.ColumnCount = 4


CASE Thisform.cFormat = "Brief"



.ColumnCount = 2


OTHERWISE 



MESSAGEBOX("Format is undefined!",48,"Bad format")



RETURN 


ENDCASE


.Column1.Header1.Caption = "Customer"


.Column1.ControlSource = "MyCursor.Customer"


.Column2.Header1.Caption = "Phone"


.Column2.ControlSource = "MyCursor.Phone"


IF Thisform.cFormat <> "Full"



RETURN


ENDIF 


.Column3.Header1.Caption = "Address"


.Column3.ControlSource = "MyCursor.Address"


.Column4.Header1.Caption = "Contact"


.Column4.ControlSource = "MyCursor.Contact"

ENDWITH
This may look like perfectly good code, but I can tell you it’s likely to cause one of those dreaded C0000005 errors. Why? Because of the two RETURN statements between the WITH and ENDWITH commands! If you RETURN after a WITH command without reaching the ENDWITH command first, you are still within the scope of that WITH…ENDWITH statement.  This is also true if you call another function while between the WITH and ENDWITH commands or even of you have an ON KEY LABEL command in effect and that key is pressed, thereby calling the other set of code.
This can really mess up VFP! To quote Doug Hennig:
“When you use WITH, VFP internally holds a reference to the WITH object. When you use ENDWITH, it releases the reference. What do you think happens if you RETURN before ENDWITH, and then later release the object? I've seen cases where the object won't release (specifically with ActiveX controls), you get dangling data sessions (the infamous "Unknown datasession"), and was the cause of C5 errors in my apps until I took the time to remove all RETURN statements within WITH...ENDWITH.”
So, given this problem, what’s the advantage of using WITH…ENDWITH? One is if you change the hierarchy of the object. If we decide to put the grid within a page of a pageframe on the form, we just have to change the WITH Thisform.MyGrid to be WITH Thisform.MyPageFrame.MyPage.MyGrid instead.

Using WITH…ENDWITH also keeps you from having to type a long hierarchy over and over.

But there’s an alternative to using WITH…ENDWITH in both of these cases. We can use a variable to store the hierarchy instead:

LOCAL loGrid

loGrid = Thisform.MyPageframe.MyPage.MyGrid
loGrid.RecordSource = "MyCursor"

DO CASE

CASE Thisform.cFormat = "Full"


loGrid.ColumnCount = 4

CASE Thisform.cFormat = "Brief"


loGrid.ColumnCount = 2

OTHERWISE 


MESSAGEBOX("Format is undefined!",48,"Bad format")

RETURN 

ENDCASE

loGrid.Column1.Header1.Caption = "Customer"

loGrid.Column1.ControlSource = "MyCursor.Customer"

loGrid.Column2.Header1.Caption = "Phone"

loGrid.Column2.ControlSource = "MyCursor.Phone"

IF Thisform.cFormat <> "Full"


RETURN

ENDIF 

loGrid.Column3.Header1.Caption = "Address"

loGrid.Column3.ControlSource = "MyCursor.Address"

loGrid.Column4.Header1.Caption = "Contact"

loGrid.Column4.ControlSource = "MyCursor.Contact"

This code is much safer. Like the WITH command, if the hierarchy changes, you only have to change the value of the variable. 

There is one advantage to WITH…ENDWITH that using a variable doesn’t have. If you use the AS clause for the WITH command as I did in the example, typing the dot will pop up Intellisense which makes it easy to select the items in the hierarchy.
Nested WITH…ENDWITH commands

It’s possible to nest WITH and ENDWITH commands. For example if you have a grid on a page of a pageframe and you want to set properties for both the page and the grid, you can use one WITH…ENDWITH to specify the page and a second WITH…ENDWITH within that specifies just the grid.
WITH Thisform.MyPageFrame.MyPage as Page


.Caption = "Customers"


.BackColor = RGB(64,128,128)


WITH .MyGrid as Grid



.RecordSource = "MyCursor"



.ColumnCount = 2



.Column1.Header1.Caption = "Customer"



.Column1.ControlSource = "MyCursor.Customer"



.Column2.Header1.Caption = "Phone"



.Column2.ControlSource = "MyCursor.Phone"


ENDWITH 

ENDWITH
I believe that WITH…ENDWITH is safe to use if you really know what you’re doing. I’ll leave it up to you to decide whether or not to use WITH…ENDWITH.
Reports

We’ve already covered part of scope regarding reports in the Data sessions topic. Now let’s look at some other aspects of reports and scope. Reports are a pretty strange breed when it comes to scope. In most cases they can see whatever the method that called them sees. That includes variables and even objects on forms. And unless the report has a private data session, that also includes data.
Reports called from an SCX-based form

For a report called from an SCX-based form, the report can see anything on the form. Period. This is because you cannot change the scope of any properties of the form or objects you drop on the form in the Form Designer.
As an example, let’s assume you have a form with two textboxes called txtDateFrom and txtDateTo. These store the date range you’d like to use for the report. In a button on the form, you have this command:
REPORT FORM MyReport TO PRINTER PROMPT PREVIEW
In the Page Header of the report, you could have a textbox that contains this expression:

"For "+DTOC(Thisform.txtDateFrom.Value)+" through "+DTOC(Thisform.txtDateTo.Value)
This would work just fine—as long as you always call the report from the form with those textboxes on it.

Likewise, you could declare and initialize a local variable in the method that calls the report and include a textbox on the report that shows that local variable.

Reports called from a PRG or VCX-based form

Most of the rules for reports called from an SCX-based form also apply to reports called from  VCX-based and PRG-based forms, except that on a VCX- or PRG-based form you have the option of changing the visibility of properties. Only properties and variables that are visible in the method calling the report are visible to the report. This is true even if the property is hidden or the variable is local. 

Keep in mind that although a report can see a hidden property of a form, objects (like command buttons) on the form cannot. This means using a hidden property on a report that’s called from the click of a command button won’t work.
#DEFINE and #INCLUDE

Commands that start with # are compiler directives. This means that anything that starts with a # is evaluated at compile time and the results are hard-coded into the executable.
#DEFINE

Using #DEFINE is a way to define constants that are compiled into the executable. In most cases the scope of #DEFINE constants is private within the item it was defined in. A constant in a PRG, SCX or VCX is private within the PRG, SCX or VCX but will not be seen if you call anything outside of the PRG, SCX or VCX, including another PRG. 
Things get really strange with forms. For either SCX-based or VCX-based, you must put any constants into an include file and use that include file in the form in order to see the constants throughout the form. (See the next section for information on include files.) If you put a #DEFINE in a method, you may be able to see it for all form methods, but you won’t be able to see it in the methods of any objects on that form.
#INCLUDE

#INCLUDE lets you specify a file of #DEFINEs or other compiler directives to be included in a class or program. Normally this file is referred to as a header file and has an extension of .H. 
The scope rules for constants brought in with #INCLUDE are the same as for #DEFINE.
When in a PRG you would simply put a #INCLUDE where you want the compiler directives to be available.

From the Class Designer or Form Designer you can specify an include file that is available anywhere in that class or form. First select “Include File” from the Class menu. See Figure 8.

Figure 8 – Option to specify an include file

A dialog to select the include file is shown. See Figure 9.

Figure 9 – Dialog to select an Include file
The scope of anything within a #INCLUDE is exactly the same as #DEFINE, with the added benefit in VCX and SCX-based forms when you use the dialog, you can see the constants from any of the objects on the form.
Executables and excluded files

When you build an executable for your app, you may opt to exclude some files. For example, you may exclude all your report files (FRX and FRT) from the executable so that you can update the reports without having to build and install an entire executable.

This is a perfectly valid technique, but you need to be aware of one limitation—once you are running something outside an EXE, you can only call back into the EXE to run code in a PRG.

As an example, say you have a screen that will need frequent updates so you’ve excluded it from the EXE. That screen calls another screen that’s included in the EXE. You test everything out on your development machine and it all runs fine. When you put it into production, you copy the new executable, plus the SCX and SCT for the excluded screen. When you run it in production, the external screen says it can’t find the screen that’s in the executable.
One thing that will not cause a problem in the above scenario is if the form uses objects derived from classes in a VCX in the executable.
There is a workaround for this problem. Remember I said that you can call back into a EXE to run something in a PRG. You can use this to call the form in the EXE indirectly. For example, you can have this simple procedure in your main program:

FUNCTION CallInternal

LPARAMETERS tcDoCmd

&tcDoCmd

ENDFUNC
You can then call CallInternal with the command you would have used directly, like this:
DO CallInternal WITH "Do Form SecondForm"
Commands that accept a scope clause

There is an additional use of the word “Scope” that’s commonly used in VFP with an entirely different meaning. In this case, I’m talking about commands that accept a scope clause, which specify which records of your data to process. Here’s a sample of some of the commands that accept a scope clause:

SCAN…ENDSCAN

CALCULATE

RECALL

COPY TO
LOCATE


SUM


TOTAL

EXPORT
DO WHILE…ENDDO
REPLACE

AVERAGE

IMPORT
REPORT FORM

DELETE

LABEL FORM
DISPLAY
You specify the scope for these commands with one of the clauses specified in the “Scope clauses” topic of the help file. These include ALL, NEXT n, REST and RECORD n. 
Note that all of these commands also accept FOR and WHILE clauses in which you specify a condition of the records to be processed. FOR and WHILE clauses are not classified as a “scope” according the help file.
While this is a completely different topic from the rest of this paper, it uses the same terminology, so I considered it worth mentioning to avoid confusion.
Appendix A – Answers to Quizzes

Exercise 1

lnVal1 displays 1

lnVal2 displays 2

pnVal3 displays 5
The attempt to display pnVa43 blows up with the error, “Variable ‘PNVAL4’ is not found.”

The reason lnVal1 displays 1 is that this is a local variable and is not affected by the different local variable of the same name used in the Function, Ex1a.

The reason lnVal2 displays 2 is similar to the answer for the display of lnVal1, but note that although lnVal2 is used in the Ex1a Function, it is not declared first, which make lnVal2 private to Ex1a. If we displayed lnVal2 in Ex1a or any function called by Ex1a, lnVal2 would display 4.

Because pnVal3 is private, it can be seen by the Ex1a Function even though it is not declared nor initialized in the Ex1a Function. Setting pnVal2 to .F. initially demonstrates how VFP does not enforce any kind of type for variables. We initialize it to a logical, and then set it to a numeric. This is perfectly legal in VFP, but is not advisable because it makes your code much harder to follow.
The situation with pnVal4 demonstrates how a private variable must be initialized in the function that declares it or it won’t be visible to that function, even if it’s initialized in a function called by the first function.
Exercise 2

This is because poMyForm is declared in Ex2CreateObj and goes out of scope once Ex2CreateObj terminates.
Exercise 3

Ex3a doesn’t work. This is because the object reference to the form is local and we call the Show method in such a way that the form is modeless. This means VFP will continue processing the code after the call to the Show method and the object reference goes out of scope immediately which releases the form too.

Ex3b works because we call the Show method with a parameter of 1, which makes the form modal. This technique is perfectly acceptable if you want a modal form. If you need the system menu or other forms active while this form is displayed, this solution isn’t acceptable.

Ex3c works because the variable storing the object reference is public. This technique is not advisable because of the use of a public variable.

Ex3d works because we’ve saved the object reference to the form as a _Screen property. The form is also modeless, which is desirable in most cases. One thing that would be better, which is missing in this example is setting the object reference back to NULL after the form is released. Unfortunately, there is no good place to do that in this example.
Exercise 4

First you will see a messagebox with “Test function in ThirdPRG” and then a messagebox with “Test function in FirstPRG”. 

This is because the first call to Test encountered is the one in SecondPRG, which sets procedure to ThirdPRG. SET PROCEDURE TO will take precedence over the calling chain.
After the call to SecondPRG from FirstPRG, FirstPRG calls Test with no qualification. Therefore, it finds the one in the current prg.

